




ДОНСКОЙ ГОСУДАРСТВЕННЫЙ ТЕХНИЧЕСКИЙ УНИВЕРСИТЕТ  
УПРАВЛЕНИЕ ДИСТАНЦИОННОГО ОБУЧЕНИЯ И ПОВЫШЕНИЯ  
КВАЛИФИКАЦИИ

Кафедра «Техническая механика»

**Практикум**  
по дисциплине

# «Нелинейные задачи строительной механики»



Авторы  
Гайджуров П.П.,  
Савельева Н.А.

Ростов-на-Дону, 2017

## Аннотация

Методические указания предназначены для студентов всех форм обучения специальности 08.05.01 «Строительство уникальных зданий и сооружений».

Приводится вывод матриц жесткости балочного и комбинированного конечных элементов. Рассмотрены вычислительные особенности формирования результирующей системы уравнений для стержневых систем. Приведены примеры анализа несущей способности рамы и ферменно-вантовой системы с учетом физической и геометрической нелинейности с использованием расчетно-вычислительного комплекса ANSYS Mechanical APDL.

## Авторы

д-р техн. наук, профессор кафедры

«Техническая механика»

Гайджуров Петр Павлович

ассистент кафедры

«Техническая механика»

Савельева Нина Александровна





## Оглавление

<b>Формирование системы уравнений стержневой системы МКЭ .....</b>	<b>4</b>
<b>Способ компактного хранения коэффициентов глобальной матрицы жесткости большой размерности ...</b>	<b>17</b>
<b>Применение расчетно-вычислительного комплекса ANSYS Mechanical APDL к анализу несущей способности стержневых конструкций с учетом физической и геометрической нелинейности .....</b>	<b>23</b>
<b>Литература .....</b>	<b>42</b>

## ФОРМИРОВАНИЕ СИСТЕМЫ УРАВНЕНИЙ СТЕРЖНЕВОЙ СИСТЕМЫ МКЭ

Систему уравнений МКЭ в форме метода перемещений представим в общепринятой форме

$$[K]\{W\} = \{P\}, \quad (1)$$

где  $[K]$  – глобальная матрица жесткости;

$\{W\}$ ,  $\{P\}$  – соответственно векторы-столбцы неизвестных узловых перемещений и заданных узловых сил конечно-элементного ансамбля.

Для формирования матрицы  $[K]$ , как правило, используют алгоритм прямого адресного включения элементных жесткостей по строкам и столбцам глобальной матрицы жесткости с последующим алгебраическим суммированием. Характерной особенностью матрицы  $[K]$  является то, что она положительно определенная, симметричная и разреженная. Последнее указывает на то, что матрица  $[K]$  содержит большое количество нулевых элементов. Поэтому в памяти компьютера достаточно хранить верхний треугольник матрицы  $[K]$ , содержащий только ненулевые элементы.

Для исключения смещений «как жесткое целое» необходимо выполнить корректировку матрицы  $[K]$  с учетом граничных условий задачи. Суть данной корректировки состоит в формальном исключении строк и столбцов матрицы  $[K]$ , соответствующих степеням свободы, в направлении которых введены связи.

Детально процесс сборки матрицы  $[K]$  можно продемонстрировать в терминах программирования. В литературе по МКЭ процесс сборки матрицы  $[K]$  называется также ансамблированием. Ниже приведен листинг программы (в терминах языка программирования Фортран) «полной» сборки матрицы  $[K]$  с комментариями, состоящей из трех вложенных циклов:

**do**  $ie = 1, ne$  ! Цикл по КЭ

# Включение матрицы жесткости  $[\bar{k}]$   $ie$ -го КЭ в  $[K]$

## Нелинейные задачи строительной механики

```

do  $i = 1, 6$  ! Внешний цикл по локальным степеням свободы
i-го КЭ
     $ia = LB[ie, i]$ 
    do  $j = 1, 6$  ! Внутренний цикл по локальным степеням сво-
боды i-го КЭ
         $ib = LB[ie, j]$ 
         $K[ia, ib] = K[ia, ib] + k[i, j]$  ! Прямое адресное включение жест-
костей i-го КЭ
    end do; end do; end do .
    
```

Здесь введены обозначения:  $ne$  – общее количество КЭ модели;  $ng$  – общее число степеней свободы модели;  $LB(ne, 6)$  – двумерный массив, предназначенный для хранения матрицы связности  $[LB]$ , устанавливающей индексную адресацию для процедуры прямого включения жесткостей  $k_{ij}$  отдельных КЭ в матрицу  $[K]$  размерностью  $ng \times ng$ . Для рассматриваемого балочного КЭ матрица жесткости имеет размерность  $6 \times 6$ . Матрицу  $[LB]$  формируем на основании принятой глобальной нумерации узлов и заданной структуры вектора-столбца узловых перемещений  $\{w\}$  стержневого КЭ. Отметим, что непосредственно перед началом сборки матрицу  $[K]$  необходимо обнулить (инициализировать).

Для учета опорных связей выполняем корректировку матрицы  $[K]$  с учетом статических граничных условий. Для этого вводим одномерный массив  $[ID]$  размерностью  $ng$ , состоящий из нулей и единиц. Причем, если на  $i$ -ю степень свободы наложена связь, то элемент массива  $ID[i] = 1$  в противном случае  $ID[i] = 0$ . В ряде случаев бывают заданными смещения узлов – это так называемые кинематические граничные условия. Для корректировки матрицы  $[K]$  с учетом кинематических граничных условий вводим массив  $[B]$  размерностью  $nw$ , где  $nw$  – число заданных узловых смещений. Для связи элементов массива  $[B]$  с элементами массива  $[ID]$  вводим информаци-

онный целочисленный массив  $[LW]$  размерностью  $nw$ . Массив  $[LW]$  предназначен для хранения номеров степеней свободы, в направлении которых заданы смещения. Таким образом, элемент  $B[j]$  является заданным узловым смещением в направлении  $LM[j]$ -ой степени свободы. Фрагмент листинга программы корректировки матрицы  $[K]$  и вектора-столбца  $\{P\}$  с учетом граничных условий приведен ниже.

```

! Корректировка матрицы [K] и вектора-столбца {P} с
! учетом граничных условий
k=1 ! Счетчик элементов массива [B]
do i = 1, ng ! Внешний цикл по глобальным степеням
! свободы
if ID[i] ≠ 0 then
do j = 1, ng ! Внутренний цикл по глобальным
! степеням свободы
if i = LM[k] then
! Корректировка элементов {P}
if j ≠ i then P[j] = P[j] - K[j,i]·B[k]; end if
if j = i then P[j] = B[k]; end if
k = k+1; end if
! Корректировка внедиагональных элементов [K]
if j ≠ i then K[i,j] = 0; K[j,i] = 0; end if
! Корректировка диагональных элементов [K]
if j = i then K[i,j] = 1; end if
end do; end if; end do .
    
```

На примере плоской рамы, показанной на рис. 1, рассмотрим порядок построения конечно-элементной сетки и формирования результирующей системы уравнений. Исходные данные: величина сосредоточенной силы, приложенной к ригелю,  $P = 1,2 \text{ кН}$ ; продольная и изгибная жесткости стоек  $EF = 9,3 \cdot 10^8 \text{ Н}$ ;  $EJ = 1,416 \cdot 10^7 \text{ Н} \cdot \text{м}^2$ . Разбивку на конечные элементы рамы выполняем так, чтобы стойкам и ригелю рамы соответствовало по одному КЭ, т. о. общее количество стержневых элементов в данном примере равно трем.




Рис. 1

Конечно-элементная схема рамы с указанием нумераций узлов, КЭ и узловых перемещений приведена на рис. 2. Узлы и степени свободы ансамбля элементов пронумерованы целыми положительными числами, начиная с единицы. Номера элементов обведены кружками. На данной схеме элемент под номером 1 является комбинированным балочным КЭ, имеющего шарнирное закрепление на конце. С каждым узлом связываем два перемещения вдоль осей  $\bar{x}$ ,  $\bar{y}$  и угол поворота относительно оси  $\bar{z}$  (направлена на наблюдателя). Параметры конечно-элементной схемы:  $ne = 3$ ;  $nr = 4$  – число узлов;  $ng = 12$ .




Рис.2




Рис. 3

В дальнейшем будем различать две системы нумерации степеней свободы – глобальную нумерацию, вводимую расчетчиком в порядке последовательного обхода узлов модели (рис. 2) и

локальную нумерацию, жестко связанную с узлами базового или комбинированного КЭ. Варианты локальной нумерации степеней свободы для  $i$ -го и  $i + 1$  – го КЭ рамы показаны на рис.3.

На основании принятой глобальной нумерации степеней свободы (рис. 2) матрица связности  $[LB]$  для конечно-элементной схемы рамы принимает вид

$$[LB] = \begin{bmatrix} 1 & 2 & 9 & 3 & 4 & 10 \\ 3 & 4 & 10 & 5 & 6 & 11 \\ 5 & 6 & 11 & 7 & 8 & 12 \end{bmatrix}.$$

(3 × 6)

Здесь первая строка соответствует принятой структуре вектора-столбца узловых перемещений первого КЭ:

$$\{\bar{w}\}_1 = \{\bar{u}_1 \ \bar{v}_1 \ \theta_1 \ \bar{u}_2 \ \bar{v}_2 \ \theta_2\}^T \text{ в осях } \{\bar{x}, \bar{y}\} \text{ и т. д.}$$

В соответствии с определением массив  $[ID]$ , предназначенный для идентификации опорных связей, имеет структуру

$$[ID] = [1 \ 1 \ 0 \ 0 \ 0 \ 0 \ 1 \ 1 \ 0 \ 0 \ 0 \ 1].$$

(1 × 12)

Матрицу жесткости первого (комбинированного) КЭ вычисляем по формуле

$$[k]_2 = \begin{bmatrix} EF/l & 0 & 0 & -EF/l & 0 & 0 \\ & 3EJ/l^3 & -3EJ/l^2 & 0 & -3EJ/l^3 & 0 \\ & & EF/l & 0 & 3EJ/l^2 & 0 \\ & & & EF/l & 0 & 0 \\ & & & & 3EJ/l^3 & 0 \\ \text{симметрично} & & & & & 1 \end{bmatrix} \quad (2);$$

Матрицу жесткости второго и третьего (базовых) КЭ вычисляем с помощью выражения



## Нелинейные задачи строительной механики

$$[k] = \begin{bmatrix} EF/l & 0 & 0 & -EF/l & 0 & 0 \\ 0 & 12EJ/l^3 & -6EJ/l^2 & 0 & -12EJ/l^3 & -6EJ/l^2 \\ 0 & -6EJ/l^2 & 4EJ/l & 0 & 6EJ/l^2 & 2EJ/l \\ -EF/l & 0 & 0 & EF/l & 0 & 0 \\ 0 & -12EJ/l^3 & 6EJ/l^2 & 0 & 12EJ/l^3 & 6EJ/l^2 \\ 0 & -6EJ/l^2 & -6EJ/l^2 & 0 & 6EJ/l^2 & 4EJ/l \end{bmatrix} \quad (3).$$

После подстановки числовых значений элементные матрицы жесткости рамы  $[k]_1$ ,  $[k]_2$ ,  $[k]_3$  в локальных осях  $\{x, y\}$  принимают значения:

$$[k]_1 = \begin{bmatrix} 1,55 \cdot 10^8 & 0 & 0 & -1,55 \cdot 10^8 & 0 & 0 \\ 0 & 1,967 \cdot 10^5 & -1,18 \cdot 10^6 & 0 & -1,967 \cdot 10^5 & 0 \\ 0 & -1,18 \cdot 10^6 & 7,08 \cdot 10^6 & 0 & 1,18 \cdot 10^6 & 0 \\ -1,55 \cdot 10^8 & 0 & 0 & 1,55 \cdot 10^8 & 0 & 0 \\ 0 & -1,967 \cdot 10^5 & 1,18 \cdot 10^6 & 0 & 1,967 \cdot 10^5 \cdot 10^6 & 0 \\ 0 & 0 & 0 & 0 & 0 & 1 \end{bmatrix};$$

$$[k]_2 = \begin{bmatrix} 4,65 \cdot 10^8 & 0 & 0 & -4,65 \cdot 10^8 & 0 & 0 \\ 0 & 5,31 \cdot 10^6 & -1,06 \cdot 10^7 & 0 & -5,31 \cdot 10^6 & -1,06 \cdot 10^7 \\ 0 & -1,06 \cdot 10^7 & 2,83 \cdot 10^7 & -5,31 \cdot 10^6 & 1,06 \cdot 10^7 & 1,42 \cdot 10^7 \\ -4,65 \cdot 10^8 & 0 & 0 & 4,65 \cdot 10^8 & 0 & 0 \\ 0 & -5,31 \cdot 10^6 & 1,06 \cdot 10^7 & 0 & 5,31 \cdot 10^6 & 1,06 \cdot 10^7 \\ 0 & -1,06 \cdot 10^7 & 1,42 \cdot 10^7 & 0 & 1,06 \cdot 10^7 & 2,83 \cdot 10^7 \end{bmatrix};$$

$$[k]_3 = \begin{bmatrix} 3,1 \cdot 10^8 & 0 & 0 & -3,1 \cdot 10^8 & 0 & 0 \\ 0 & 6,293 \cdot 10^6 & -9,44 \cdot 10^6 & 0 & -6,293 \cdot 10^6 & -9,44 \cdot 10^6 \\ 0 & -9,44 \cdot 10^6 & 1,888 \cdot 10^7 & 0 & 9,44 \cdot 10^6 & 9,44 \cdot 10^6 \\ -3,1 \cdot 10^8 & 0 & 0 & 3,1 \cdot 10^8 & 0 & 0 \\ 0 & -6,293 \cdot 10^6 & 9,44 \cdot 10^6 & 0 & 6,293 \cdot 10^6 & 9,44 \cdot 10^6 \\ 0 & -9,44 \cdot 10^6 & 9,44 \cdot 10^6 & 0 & 9,44 \cdot 10^6 & 1,888 \cdot 10^7 \end{bmatrix}.$$

Матрицы «поворота»  $[c]_1$ ,  $[c]_2$ ,  $[c]_3$  для КЭ данной

рамы, вычисляются по формуле

$$[c] = \begin{bmatrix} \cos \varphi & -\sin \varphi & 0 & 0 & 0 & 0 \\ \sin \varphi & \cos \varphi & 0 & 0 & 0 & 0 \\ 0 & 0 & 1 & 0 & 0 & 0 \\ 0 & 0 & 0 & \cos \varphi & -\sin \varphi & 0 \\ 0 & 0 & 0 & \sin \varphi & \cos \varphi & 0 \\ 0 & 0 & 0 & 0 & 0 & 1 \end{bmatrix} \quad (4),$$

принимают значения:

$$[c]_1 = \begin{bmatrix} 0 & -1 & 0 & 0 & 0 & 0 \\ 1 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 1 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & -1 & 0 \\ 0 & 0 & 0 & 1 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 1 \end{bmatrix}; \quad [c]_2 = \begin{bmatrix} 1 & 0 & 0 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 & 0 & 0 \\ 0 & 0 & 1 & 0 & 0 & 0 \\ 0 & 0 & 0 & 1 & 0 & 0 \\ 0 & 0 & 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 0 & 0 & 1 \end{bmatrix};$$

$$[c]_3 = \begin{bmatrix} 0 & -1 & 0 & 0 & 0 & 0 \\ 1 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 1 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & -1 & 0 \\ 0 & 0 & 0 & 1 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 1 \end{bmatrix}.$$

С помощью выражения

$$[\bar{k}] = [c]^T [k] [c] \quad (5)$$

выполним преобразование матриц  $[k]_1, [k]_2, [k]_3$ , относящихся к местной системе осей  $\{x, y\}$ , в матрицы  $[\bar{k}]_1$

,  $[\bar{k}]_2$ ,  $[\bar{k}]_3$ , относящиеся к глобальным осям  $\{\bar{x}, \bar{y}\}$ . Как видно из структуры матриц  $[c]_1$ ,  $[c]_2$ ,  $[c]_3$  данное преобразование затронет только матрицы  $[k]_1$ ,  $[k]_3$ , т. к. матрица  $[c]_2$  является единичной. В результате получим:

$$[\bar{k}]_1 = \begin{bmatrix} 1,967 \cdot 10^5 & 0 & 1,18 \cdot 10^6 & -1,967 \cdot 10^5 & 0 & 0 \\ 0 & 1,55 \cdot 10^8 & 0 & 0 & -1,55 \cdot 10^8 & 0 \\ 1,18 \cdot 10^6 & 0 & 7,08 \cdot 10^6 & -1,18 \cdot 10^6 & 0 & 0 \\ -1,967 \cdot 10^5 & 0 & -1,18 \cdot 10^6 & 1,967 \cdot 10^5 & 0 & 0 \\ 0 & -1,55 \cdot 10^8 & 0 & 0 & 1,55 \cdot 10^8 & 0 \\ 0 & 0 & 0 & 0 & 0 & 1 \end{bmatrix};$$

$$[\bar{k}]_2 = [k]_2;$$

$$[\bar{k}]_3 = \begin{bmatrix} 6,293 \cdot 10^6 & 0 & -9,44 \cdot 10^6 & -6,293 \cdot 10^6 & 0 & -9,44 \cdot 10^6 \\ 0 & 3,1 \cdot 10^8 & 0 & 0 & -3,1 \cdot 10^8 & 0 \\ -9,44 \cdot 10^6 & 0 & 1,888 \cdot 10^7 & 9,44 \cdot 10^6 & 0 & 9,44 \cdot 10^6 \\ -6,293 \cdot 10^6 & 0 & 9,44 \cdot 10^6 & 6,293 \cdot 10^6 & 0 & 9,44 \cdot 10^6 \\ 0 & -3,1 \cdot 10^8 & 0 & 0 & 3,1 \cdot 10^8 & 0 \\ -9,44 \cdot 10^6 & 0 & 9,44 \cdot 10^6 & 9,44 \cdot 10^6 & 0 & 1,888 \cdot 10^7 \end{bmatrix}.$$

После элементной сборки глобальная матрица жесткости рамы принимает форму:

## Нелинейные задачи строительной механики

$$[K] = \begin{bmatrix}
 1,967 \cdot 10^5 & 0 & -1,967 \cdot 10^5 & 0 & 0 & 0 \\
 0 & 1,55 \cdot 10^8 & 0 & -1,55 \cdot 10^8 & 0 & 0 \\
 -1,967 \cdot 10^5 & 0 & 4,652 \cdot 10^8 & 0 & 4,65 \cdot 10^8 & 0 \\
 0 & -1,55 \cdot 10^8 & 0 & 1,603 \cdot 10^8 & 0 & -5,31 \cdot 10^6 \\
 0 & 0 & -4,65 \cdot 10^8 & 0 & 4,713 \cdot 10^8 & 0 \\
 0 & 0 & 0 & -5,31 \cdot 10^6 & 0 & 3,153 \cdot 10^8 \\
 0 & 0 & 0 & 0 & -6,293 \cdot 10^6 & 0 \\
 0 & 0 & 0 & 0 & 0 & -3,1 \cdot 10^8 \\
 1,18 \cdot 10^6 & 0 & -1,18 \cdot 10^6 & 0 & 0 & 0 \\
 0 & 0 & 0 & -1,062 \cdot 10^7 & 0 & 1,062 \cdot 10^7 \\
 0 & 0 & 0 & -1,062 \cdot 10^7 & -9,44 \cdot 10^6 & 1,062 \cdot 10^7 \\
 0 & 0 & 0 & 0 & -9,44 \cdot 10^6 & 0 \\
 \\
 0 & 0 & 1,18 \cdot 10^6 & 0 & 0 & 0 \\
 0 & 0 & 0 & 0 & 0 & 0 \\
 0 & 0 & -1,18 \cdot 10^6 & 0 & 0 & 0 \\
 0 & 0 & 0 & -1,062 \cdot 10^7 & -1,062 \cdot 10^7 & 0 \\
 -6,293 \cdot 10^6 & 0 & 0 & 0 & -9,44 \cdot 10^6 & -9,44 \cdot 10^6 \\
 0 & -3,1 \cdot 10^8 & 0 & 1,062 \cdot 10^7 & 1,062 \cdot 10^7 & 0 \\
 \dots & 6,293 \cdot 10^6 & 0 & 0 & 9,44 \cdot 10^6 & 9,44 \cdot 10^6 \\
 0 & 3,1 \cdot 10^8 & 0 & 0 & 0 & 0 \\
 0 & 0 & 7,08 \cdot 10^6 & 0 & 0 & 0 \\
 0 & 0 & 0 & 2,832 \cdot 10^7 & 1,416 \cdot 10^7 & 0 \\
 9,44 \cdot 10^6 & 0 & 0 & 1,416 \cdot 10^7 & 4,72 \cdot 10^7 & 9,44 \cdot 10^6 \\
 9,44 \cdot 10^6 & 0 & 0 & 0 & 9,44 \cdot 10^6 & 1,888 \cdot 10^7
 \end{bmatrix}$$

Для того, чтобы исключить смещение рамы «как жесткое целое» необходимо выполнить ее корректировку с учетом граничных условий задачи. В результате корректировки матрица  $[K]$  преобразуется к виду

## Нелинейные задачи строительной механики

$$[K] = \begin{matrix} (12 \times 12) \\ \left[ \begin{array}{cccccc} 1 & 0 & 0 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 & 0 & 0 \\ 0 & 0 & 4,652 \cdot 10^8 & 0 & 4,65 \cdot 10^8 & 0 \\ 0 & 0 & 0 & 1,603 \cdot 10^8 & 0 & -5,31 \cdot 10^6 \\ 0 & 0 & -4,65 \cdot 10^8 & 0 & 4,713 \cdot 10^8 & 0 \\ 0 & 0 & 0 & -5,31 \cdot 10^6 & 0 & 3,153 \cdot 10^8 \\ 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & -1,18 \cdot 10^6 & 0 & 0 & 0 \\ 0 & 0 & 0 & -1,062 \cdot 10^7 & 0 & 1,062 \cdot 10^7 \\ 0 & 0 & 0 & -1,062 \cdot 10^7 & -9,44 \cdot 10^6 & 1,062 \cdot 10^7 \\ 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & -1,18 \cdot 10^6 & 0 & 0 & 0 \\ 0 & 0 & 0 & -1,062 \cdot 10^7 & -1,062 \cdot 10^7 & 0 \\ 0 & 0 & 0 & 0 & -9,44 \cdot 10^6 & 0 \\ \dots & 0 & 0 & 1,062 \cdot 10^7 & 1,062 \cdot 10^7 & 0 \\ 1 & 0 & 0 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 & 0 & 0 \\ 0 & 0 & 7,08 \cdot 10^6 & 0 & 0 & 0 \\ 0 & 0 & 0 & 2,832 \cdot 10^7 & 1,416 \cdot 10^7 & 0 \\ 0 & 0 & 0 & 1,416 \cdot 10^7 & 4,72 \cdot 10^7 & 0 \\ 0 & 0 & 0 & 0 & 0 & 1 \end{array} \right] \end{matrix} .$$

Решив систему уравнений (1) с откорректированной матрицей  $[K]$  и вектором-столбцом узловых сил

$$\{P\} = \{0 \ 0 \ 1200 \ 0 \ 0 \ 0 \ 0 \ 0 \ 0 \ 0 \ 0 \ 0\}^T,$$

получим вектор-столбец глобальных узловых перемещений:

$$\{W\} = \{0 \quad 0 \quad 2,98 \cdot 10^{-4} \quad 2,37 \cdot 10^{-6} \quad 2,96 \cdot 10^{-4} \quad -1,18 \cdot 10^{-6} \dots$$

$$\dots 0 \quad 0 \quad 4,97 \cdot 10^{-5} \quad -3,37 \cdot 10^{-5} \quad 7,01 \cdot 10^{-5} \quad 0\}^T.$$

Здесь линейные перемещения измеряются в метрах, а угловые – в радианах. Визуализация деформированного состояния рамы представлена на рис. 4.




Рис. 4

Отметим, что в учебных целях для решения результирующей системы уравнений (1) может быть использована среда компьютерной математики систем Matlab, Mathcad, Maple.

По найденным узловым перемещениям  $\{W\}$ , используя матрицу  $[LB]$  и формулу  $\{w\} = [c]^T \{\bar{w}\}$ , определим узловые перемещения для каждого КЭ. Узловые усилия в местных осях находим с помощью выражения  $\{p\} = [k]\{w\}$ , в которое для комбинированных КЭ вместо матрицы (3) подставляем матрицу (2).

Векторы-столбцы узловых сил КЭ в местных осях принимают значения:

$$\{p\}_1 = \{-367,3 \quad 0 \quad 1 \cdot 10^{-7} \quad 367,3 \quad 0 \quad -0,337 \cdot 10^{-4}\}^T;$$

$$\{p\}_2 = \{1200 \quad -367,3 \quad 0,342 \cdot 10^{-4} \quad -1200 \quad 367,3 \quad 1469\}^T;$$

$$\{p\}_3 = \{367,3 \quad 1200 \quad 1469 \quad -367,3 \quad -1200 \quad -2131\}^T,$$

где узловые усилия и моменты измеряются соответственно в Н и Н·м. Естественно малыми величинами узловых моментов ( $1 \cdot 10^{-7}$ ,  $-0,337 \cdot 10^{-4}$ ,  $0,342 \cdot 10^{-4}$ ) можно пренебречь, считая их равными нулю.

Эпюры изгибающих моментов  $M$ , поперечных  $Q$  и продольных  $N$  сил, построенные на основании значений  $\{p\}_1$ ,  $\{p\}_2$ ,  $\{p\}_3$ , приведены на рис. 5. Здесь ординаты  $M$ ,  $Q$ ,  $N$  отложены в соответствии с общепринятым в строительной механике правилом знаков.




Рис. 5




Рис. 6




Рис. 7

В заключении отметим, что моделировать неопорное шарнирное соединение стержней с помощью комбинированных КЭ можно по конечно-элементным схемам, приведенным на рис. 6 и 7.



## СПОСОБ КОМПАКТНОГО ХРАНЕНИЯ КОЭФФИЦИЕНТОВ ГЛОБАЛЬНОЙ МАТРИЦЫ ЖЕСТКОСТИ БОЛЬШОЙ РАЗМЕРНОСТИ

Как известно в МКЭ структура глобальной матрицы жесткости является симметричной, разреженной и существенно зависит от порядка нумерации узлов модели. Для хранения элементов глобальной матрицы жесткости  $[K]$  большой размерности разработаны специальные методы. Рассмотрим способ представления симметричной разреженной матрицы в виде одномерного массива в формате *RR(C)O* («*Row-wise Representation Complete and Ordered*» – пер. с англ. строчное представление, полное и упорядоченное). Характерной особенностью построчного хранения является представление данных о двумерном массиве в виде четырех одномерных массивов: двух целочисленных и двух вещественных. Для хранения построчных значений ненулевых недиагональных элементов матрицы  $[K]$  вводится массив  $an$ ; элементы, расположенные на главной диагонали хранятся в массиве  $ad$ . С целью позиционирования недиагональных элементов вводятся целочисленные массивы указателей  $ja$  и  $ia$ . Причем в массиве  $ja$  хранятся индексы столбцов ненулевых элементов ГМЖ, а в массиве  $ia$  – указатели, отмечающие позиции массивов  $an$  и  $ja$ , с которых начинается описание очередной строки ГМЖ.

Наглядное представление о строчном формате можно получить с помощью следующего примера:

$$[A] = \begin{matrix} (5 \times 5) & \begin{bmatrix} 10 & 0 & 1 & 0 & 2 \\ 0 & 20 & 0 & 0 & 3 \\ 1 & 0 & 30 & 4 & 0 \\ 0 & 0 & 4 & 40 & 5 \\ 2 & 3 & 0 & 5 & 50 \end{bmatrix} \end{matrix} .$$

Матрица  $[A]$  в формате *RR(C)O* имеет вид табл. 1.

Дополнительные элементы в массиве  $ia$  (позиция 5)

содержат указатель первой свободной позиции в массиве  $ja$  и соответственно в массиве  $an$ .

Данное представление матрицы  $[A]$  является упорядоченным.

Таблица 1

Позиция	1	2	3	4	5
$ia$	1	3	4	5	5
$ja$	3	5	5	4	5
$an$	1	2	3	4	5
$ad$	10	20	30	40	50

Массивы  $ia$  и  $ja$  представляют так называемый портрет матрицы  $[A]$ , задаваемый как множества списков смежности, ассоциированного с  $[A]$  графа. Как правило, в алгоритме конечно-элементной сборки и решения результирующей системы алгебраических уравнений предусматривается разделение вычислительного процесса на символьную и численную процедуры. При этом на этапе символьной обработки информации формируется портрет ГМЖ, а на этапе численной обработки происходит вычисление или преобразование значений массивов  $an$  и  $ad$ .

При конечно-элементной сборке используется разреженный строчный формат  $RR(C)U$  («*Row-wise Representation Complete and Unordered*» – пер. с англ. строчное представление, полное, но неупорядоченное). Формат  $RR(C)U$  отличается от формата  $RR(C)O$  тем, что в данном случае соблюдается упорядоченность строк матрицы  $[K]$ , однако внутри каждой строки элементы могут храниться в произвольном порядке.

В расчетной практике при дискретизации стержневой системы на конечные элементы, как правило, каждый стержень разбивают на несколько КЭ, используя равномерную сетку (постоянный шаг разбивки). Данный прием с одной стороны позволяет в наглядной форме представить результаты расчета внутренних усилий, а с другой стороны является обязательным при анализе устойчивости и динамических расчетах стержневых конструкций. При этом количество КЭ, на которое разбивается каждый стержень, может быть различным и определяется путем про-

ведения серии численных расчетов.

В качестве примера, демонстрирующего способ компактного хранения матрицы  $[K]$ , рассмотрим раму, показанную на рис. 8.

Параметры конечно-элементной схемы, представленной на рис. 9, следующие:  $n_e=13$ ;  $n_r=14$ ;  $n_g=42$ . Примеры локальной нумерации для  $i$ -го и  $i+1$ -го КЭ показаны на рис. 9б).




Рис. 8




Рис. 9

Введем матрицу инцидентности  $[I]$  (связности), устанавливающую однозначное соответствие между строками и столбцами глобальной и локальной матриц жесткости. Матрица  $[I]$  имеет размерность  $n_e \times n_g$  и является булевой матрицей, т. е. состоит из нулей и единиц. Причем  $I_{ij} = 1$ , если номер  $j$  глобальной степени свободы принадлежит элементу  $i$ . В противном случае  $I_{ij} = 0$ . Для рассматриваемого примера портрет (структура в символическом виде) матрицы  $[I]$  имеет вид, показанный на рис. 10.

## Нелинейные задачи строительной механики




Рис. 10

На практике построение матрицы  $[I]$  полностью автоматизировано. Как правило, вводится топологическая информация о нумерации узлов ансамбля элементов, на основании которой формируется матрица  $[I]$ .

Формально процесс конечно-элементной сборки можно представить в виде символической процедуры произведения матриц  $[I]^T [I]$ . В результате получаем портрет глобальной матрицы жесткости  $[K]$  (рис. 11).

$$(42 \times 42)$$

## Нелинейные задачи строительной механики




Рис. 11

Отметим, что при организации вычислительного процесса процедура символической сборки ( $[I]^T [I]$ ) предшествует процедуре численного включения элементной матрицы жесткости  $[\bar{k}]$  и вектора сил  $\{\bar{p}\}$  в глобальную матрицу жесткости  $[K]$  и соответствующий глобальный вектор узловых сил  $\{P\}$ .

## ПРИМЕНЕНИЕ РАСЧЕТНО-ВЫЧИСЛИТЕЛЬНОГО КОМПЛЕКСА ANSYS MECHANICAL APDL К АНАЛИЗУ НЕСУЩЕЙ СПОСОБНОСТИ СТЕРЖНЕВЫХ КОНСТРУКЦИЙ С УЧЕТОМ ФИЗИЧЕСКОЙ И ГЕОМЕТРИЧЕСКОЙ НЕЛИНЕЙНОСТИ

Расчетно-вычислительный комплекс ANSYS Mechanical APDL позволяет выполнять инженерные расчеты стержневых систем с учетом как геометрической, так и физической нелинейностей. Данный вид расчетов востребован при анализе несущей способности строительных конструкций при запроектных внешних воздействиях, а также при анализе сценария прогрессирующего (лавинообразного) обрушения здания или сооружения. Рассмотрим числовые примеры.

**Пример 1.** Выполним упругопластический расчет плоской статически неопределимой рамы, показанной на рис. 12.




Рис. 12

Исходные данные:  $l_1 = 2$  м;  $l_2 = 4$  м;  $l_3 = 3$  м;  $b_1 = 0,3$  м;  $h_1 = 0,2$  м;  $b_2 = 0,2$  м;  $h_2 = 0,36$  м;  $k = 0,6$ .

Принимаем, что стойки и ригель рамы выполнены из идеально упругопластического материала одинаково работающего при растяжении и сжатии (рис. 13). Предел упругости материала

$$\sigma_T = 285 \text{ МПа.}$$

Стержни рамы моделируем 2-х узловыми балочными КЭ ти-

па BEAM23 с тремя степенями свободы в узле (две линейные и одна угловая). Данные КЭ позволяют выполнять расчеты плоских рамных конструкций с учетом упругопластического деформирования.




Рис. 13

Для формирования геометрии, назначения жесткостей стержней, задания граничных условий, нагружения рамы, построения конечно-элементной сетки и численного решения результирующей нелинейной системы уравнений используем встроенный в ANSYS язык параметрического программирования APDL, который по своему синтаксису напоминает Фортран. Ниже приведен листинг соответствующего макроса, составленный с помощью текстового редактора «Блокнот».

```

FINISH
/CLEAR,NOSTART ! очистка от предыдущих данных
/PREP7 ! Вход в препроцессор
!-----
!Ввод параметров исходных данных:
!Геометрические параметры, м
l1=2
l2=4
l3=3
b1=0.3
h1=0.2
b2=0.2
h2=0.36
!Константы материала
E=2e11 ! Модуль упругости, Н/м^2
nu=0.28 ! Коэффициент Пуассона
sigT=285e06 ! Предел упругости, Н/м^2
    
```





## Нелинейные задачи строительной механики

```
Ek=0.01*E ! модуль упрочнения, Н/м^2
!Коэффициент нагрузки
!-----
!Параметры нагружения
k=0.6
N_STEP=50 ! Число шагов нагружения
P_MAX=1750000 ! Максимальное значение параметра
нагружения, Н
!-----
!Вычисление геометрических характеристик сечений
!Стойки
F1=b1*h1 ! Площадь сечения №1, м^2
J1=(b1*h1**3)/12 ! Осевой момент инерции сечения
№1, м^4
!Ригель
F2=b2*h2 ! Площадь сечения №2, м^2
J2=(b2*h2**3)/12 ! Осевой момент инерции сечения
№2, м^4
!-----
!Геометрия рамы. Ввод координат ключевых точек, м
K,1,0,0,0 ! Точка 1
K,2,0,13,0 ! Точка 2
K,3,11,13,0 ! Точка 3; координата точки приложения
силы P
K,4,11+12,13,0 ! Точка точка 4
K,5,11+12,0,0 ! Точка точка 5
!-----
!Соединение ключевых точек линиями
LSTR,1,2 ! Создание линии между точками 1 и 2
LSTR,2,3 ! Создание линии между точками 2 и 3
LSTR,3,4 ! Создание линии между точками 3 и 4
LSTR,4,5 ! Создание линии между точками 4 и 5
!-----
!Выбор конечного элемента тип BEAM23
!(2D балка для упругопластического анализа)
ET,1,BEAM23
!Параметры BEAM23
KEYOPT,1,2,1
KEYOPT,1,4,1
KEYOPT,1,6,0
!-----
!Ввод геометрических характеристик стержней рамы
!Стойки
R,1,F1,J1,h1
!Ригель
```



## Нелинейные задачи строительной механики

```
R, 2, F2, J2, h2
!-----
!Назначение механических констант материала
UIMP,1,EX,,,E ! Модуль упругости материала
UIMP,1,PRXY,,,nu ! Коэффициент Пуассона
!Параметры билинейной диаграммы деформирования
TB,BISO,1,1,2,
TBTEMP,0
TBDATA,,sigT,Ek,,,,
!-----
!Задание граничных условий (связей)
!Покоординатное выделение точки левой опоры
KSEL,S,LOC,X,0
KSEL,R,LOC,Y,0
KSEL,R,LOC,Z,0
DK,ALL,,,,,ALL,,,, ! Жесткая заделка в выделенной
точке
!Покоординатное выделение точки правой опоры
KSEL,S,LOC,X,l1+l2
KSEL,R,LOC,Y,0
KSEL,R,LOC,Z,0
DK,ALL,,,,,ALL,,,, ! Жесткая заделка в выделенной
точке
ALLSEL,ALL ! Выделить все объекты
!-----
!Присвоение атрибутов конечных элементов линиям
LSEL,S,LINE,,1 ! Выделение линии L1
LATT,1,1,1,0 ! Материал №1; константы real 1; эле-
мент BEAM3
LESIZE,1,1 ! Разбиение линии L1 на конечные элемен-
ты с шагом 1м
LSEL,S,LINE,,2 ! Выделение линии L1
LATT,1,2,1,0 ! Материал №1; константы real 2; эле-
мент BEAM3
LESIZE,2,1 ! Разбиение линии L2 на конечные элемен-
ты с шагом 1м
LSEL,S,LINE,,3 ! Выделение линии L3
LATT,1,2,1,0 ! Материал №1; константы real 2; эле-
мент BEAM3
LESIZE,3,1 ! Разбиение линии L3 на конечные элемен-
ты с шагом 1м
LSEL,S,LINE,,4 ! Выделение линии L4
LATT,1,1,1,0 ! Материал №1; константы real 1; эле-
мент BEAM3
LESIZE,4,1 ! Разбиение линии L4 на конечные элемен-
```



## Нелинейные задачи строительной механики

```
ты с шагом 1м
ALLSEL,ALL ! Выделить все объекты
LMESH,ALL ! Построение сетки конечных элементов
FINISH ! Выход из препроцессора
!-----
/SOLU ! Вход в решатель
!
ANTYPE,0 ! Статический расчет
NLGEOM,1 ! Учет больших перемещений
OUTPR,BASIC,ALL, ! Вывод результатов на каждом шаге
OUTRES,ALL,ALL, ! Контроль решения на каждом шаге
*DO,I_STEP,1,N_STEP,1 ! Цикл по шагам нагружения
P_STEP=P_MAX*(I_STEP/N_STEP) ! Текущее значение па-
раметра нагружения
TIME,I_STEP ! Задание параметра очередного шага
нагружения
!Покоординатное выделение точки приложения силы P
KSEL,S,LOC,X,11
KSEL,R,LOC,Y,13
KSEL,R,LOC,Z,0
FK,ALL,FY,-P_STEP ! Приложение силы P_STEP в выде-
ленной точке по направлению Y
!Покоординатное выделение точки приложения силы kP
KSEL,S,LOC,X,0
KSEL,R,LOC,Y,13
KSEL,R,LOC,Z,0
FK,ALL,FX,k*P_STEP ! Приложение силы k*P_STEP в вы-
деленной точке по направлению X
ALLSEL,ALL ! Выделить все объекты
SOLVE ! Решение системы уравнений на шаге I_STEP
*ENDDO ! Конец цикла по I_STEP
FINISH ! Выход из нелинейного решателя
!-----
```

Процесс нагружения рамы организован в виде шаговой процедуры. Запуск макроса на выполнение осуществляется путем его копирования в буфер обмена с последующей вставкой в командную строку ANSYS и нажатия клавиши «Enter» (рис.14).

Нелинейные задачи строительной механики




Рис. 14

При работе процессора решения нелинейной системы уравнений (/SOLU) на экране монитора появляется окно визуализации параметров итерационного процесса на текущем шаге нагружения (рис. 15). Следует отметить, что нелинейный решатель комплекса ANSYS при превышении предельного значения параметра нагружения «аварийно» завершает работу. Поэтому для корректного задания величины параметра нагружения P\_MAX требуется серия вычислительных экспериментов.

Нелинейные задачи строительной механики





Рис. 15

На рис. 16 и 17 представлены результаты расчета на последнем шаге нагружения соответственно в виде картин деформированного состояния рамы и распределения в стержнях рамы интенсивности пластических деформаций  $\epsilon$  и  $p$ .



```

ANSYS 14.5.7
DISPLACEMENT
STEP=50
SUB =38
TIME=50
PowerGraphics
EFACET=1
AVRES=Mat
DMX =.107379
    
```

Рис. 16

На рис. 17 стрелка указывает на область, в которой образовался пластический шарнир. В данном случае эта область принадлежит элементу стойки, примыкающему к ригелю.




Рис. 17

Графики горизонтального  $W_x$  и вертикального  $W_y$  перемещений точки приложения сосредоточенной силы  $P$  в зависимости от номера шага нагружения  $i\_step$  представлены на рис. 18 и 19.




Рис. 18





Рис. 19

Как следует из приведенных графиков, линейно упругая стадия работы рамы заканчивается на шаге  $i\_step = 30$ . Далее вследствие пластического отклика материала наблюдается нелинейная зависимость между наблюдаемыми перемещениями и нагрузкой. На последнем шаге нагружения  $i\_step = 50$  происходит, так называемый, «вертикальный срыв решения», что указывает на исчерпание несущей способности рамы. Эпюра изгибающих моментов  $M$ , соответствующая последнему шагу нагружения приведена на рис. 20.


 $M, Н\cdot м$ 

A	=-899573
B	=-538059
C	=-176545
D	=184969
E	=546482
F	=907996
G	=.127E+07
H	=.163E+07
I	=.199E+07

Рис. 20

Сопоставляя эпюру  $M$  с картиной распределения  $\varepsilon_{иp}$  (рис. 17), приходим к выводу, что сечение, в котором возникает экстремальный изгибающий момент ( $0,199 \cdot 10^7 \text{ Н}\cdot\text{м}$ ), не совпадает с положением пластического шарнира. Это объясняется тем, что жесткость ригеля выше жесткости стоек.

В заключении данного примера приведем значение предельной нагрузки, вычисленное по методу предельного равновесия  $P_{\text{пр}} = 1910 \text{ кН}$ . Для сравнения МКЭ дает значение предельной нагрузки равное  $1750 \text{ кН}$ , что почти на 10% меньше.

**Пример 2.** Выполнить расчет плоской ферменно-вантовой конструкции, показанной на рис. 21, на прогрессирующее разрушение, обусловленное выходом из строя одного из предварительно растянутых тросов.

Исходные данные: габаритные размеры конструкции  $l = 12 \text{ м}$ ;  $h = 0,65 \text{ м}$ ;  $y_0 = 0,5 \cdot l$ . Материал фермы и тросов – сталь.

Ферма выполнена из холодно-катанной трубы: несущие пояса – труба диаметром  $100 \text{ мм}$ , толщина стенки  $t = 5 \text{ мм}$ ; стойки и раскосы – труба диаметром  $75 \text{ мм}$ , толщина стенки  $t = 3 \text{ мм}$ . Диаметр тросовых растяжек (вант)  $27 \text{ мм}$ .

В качестве нагрузки учитываем только собственный вес фермы.

Стержни моделируем ферменными КЭ типа LINK1 с двумя линейными степенями свободы в узле. Данный тип КЭ позволяет учитывать начальную деформацию стержня.




Рис. 21



## Нелинейные задачи строительной механики

Ниже приведен листинг макроса, с помощью которого выполняется построение геометрии данной стержневой системы, осуществляется разбивка стержней на ферменные КЭ, накладываются связи, и задается нагрузка (собственный вес фермы).

Сценарии прогрессирующего разрушения представим в виде отключения (обрыва) одной из тросовых растяжек. При этом конструкция, находившаяся в статическом состоянии, после обрыва троса будет совершать вынужденные затухающие колебания. Для отключения тросовой растяжки в программе макроса предусмотрен флажок del.

```

FINISH
/CLEAR,NOSTART ! очистка от предыдущих данных
!-----
/PREP7 ! Вход в препроцессор
!-----
!Ввод параметров исходных данных:
!Геометрические параметры
l=12 ! Длина фермы, м
h=0.65 ! Высота фермы, м
n=12 ! Число прогонов фермы (число прогонов четное
и кратное значению l)
y0=0.5*l ! Ордината верхней точки подвеса тросов
d1=0.1 ! Диаметр трубы несущего пояса фермы, м
t1=0.005 ! Толщина стенки трубы несущего пояса фер-
мы, м
d2=0.075 ! Диаметр трубы стойки и раскоса фермы, м
t2=0.003 ! Толщина стенки трубы стойки и раскоса
фермы, м
d3=0.027 ! Диаметр троса, м
!-----
!Геометрические характеристики стержней
d0=d1-2*t1 ! Внутренний диаметр трубы несущего поя-
са
f1=3.14*(d1**2-d0**2)/4 ! Площадь сечения несущего
пояса, м^2
d0=d2-2*t2 ! Внутренний диаметр трубы стойки и рас-
коса
f2=3.14*(d2**2-d0**2)/4 ! Площадь сечения стойки и
раскоса, м^2
f3=3.14*d3**2/4 ! Площадь поперечного сечения тро-
совой растяжки, м^2
!-----
!Константы материала (сталь)
    
```

## Нелинейные задачи строительной механики

```

E=2e11 ! Модуль упругости, Н/м^2
nu=0.28 ! Коэффициент Пуассона
ro=7800 ! Плотность материала, кг/м^3
!-----
eps1=-0.008 ! Деформация наружной растяжки от пред-
варительного натяжения
eps2=-0.003 ! Деформация внутренней растяжки от
предварительного натяжения
!Параметры конечно-элементной сетки
n1=8 ! Число делений на конечные элементы наружного
троса
n2=6 ! Число делений на конечные элементы внутрен-
него троса
!-----
!Флажок отключения тросовой растяжки
!del=0 ! Расчет без отключения
del=1 ! Отключается наружная растяжка
!del=2 ! Отключается внутренняя растяжка
!-----
!Выбор конечного элемента тип LINK1 (2D ферменный
стержень)
ET,1,LINK1
!-----
!Ввод геометрических характеристик стержней фермы и
растяжек
!Несущие пояса
R,1,f1
!Стойки и раскосы
R,2,f2
!Тросовая растяжка наружная
R,3,f3,eps1
!Тросовая растяжка внутренняя
R,4,f3,eps2
!-----
!Назначение механических констант материала №1 для
фермы
UIMP,1,EX,,,E ! Модуль упругости материала
UIMP,1,PRXY,,,nu ! Коэффициент Пуассона
UIMP,1,DENS,,,ro ! Плотность материала
!Назначение механических констант материала №2 для
тросов
UIMP,2,EX,,,E ! Модуль упругости материала
UIMP,2,PRXY,,,nu ! Коэффициент Пуассона
UIMP,2,DENS,,,0.0001*ro ! Плотность материала
!-----
    
```

## Нелинейные задачи строительной механики

```

!Геометрия рамы. Ввод координат ключевых точек, m
step=1/n ! Длина прогона фермы
K,1,0,0,0 ! Точка 1
K,2,0,h,0 ! Точка 2
KGEN,n+1,1,2,1,step ! Масштабирование точек 1 и 2
вдоль оси X
!Определение наибольшего номера ключевой точки
kp_max
*get,kp_max,kp,,num,max
K,kp_max+1,0,y0,0 ! Точка верхнего подвеса тросов
!-----
!Соединение ключевых точек линиями в заданной по-
следовательности
k1=1
k2=2
*do,i,1,n+1 ! Цикл по стержням стоек
LSTR,k1,k2 ! Создание линии между точками k1 и k2
k1=k1+2
k2=k2+2
*enddo ! i
k1=1
k2=3
*do,i,1,2*n ! Цикл по стержням верхнего и нижнего
поясов
LSTR,k1,k2 ! Создание линии между точками k1 и k2
k1=k1+1
k2=k2+1
*enddo ! i
k1=1
k2=4
k3=4
k4=5
*do,i,1,n/2 ! Цикл по стержням раскосов
LSTR,k1,k2 ! Создание линии между точками k1 и k2
LSTR,k3,k4 ! Создание линии между точками k3 и k4
k1=k1+4
k2=k2+4
k3=k3+4
k4=k4+4
*enddo ! i
!Соединение точек подвеса тросовых растяжек
k1=kp(0,y0,0)
k2=kp(1,h,0)
LSTR,k1,k2 ! Линия наружного троса
k2=kp(1/2,h,0)
    
```



## Нелинейные задачи строительной механики

```
LSTR,k1,k2 ! Линия внутреннего троса
!-----
!Задание граничных условий (связей)
!Покоординатное выделение точки с координатами x=0,
y=0
KSEL,S,LOC,X,0
KSEL,R,LOC,Y,0
KSEL,R,LOC,Z,0
DK,ALL,,,,,UX,UY,,,, ! Шарнирно-неподвижная опора
!Покоординатное выделение точки верхнего подвеса
тросов
KSEL,S,LOC,X,0
KSEL,R,LOC,Y,y0
KSEL,R,LOC,Z,0
DK,ALL,,,,,UX,UY,,,, ! Шарнирно-неподвижная опора
ALLSEL,ALL ! Выделить все объекты
!-----
!Присвоение атрибутов конечных элементов линиям
n_line=1 ! Счетчик номеров линий фермы
*do,i,1,n+1 ! Цикл по стержням стойек
LSEL,S,LINE,,n_line ! Выделение линии № n_line
LATT,1,2,1,0 ! Материал №1; константы real 2; эле-
мент LINK1
LESIZE,n_line,,,1 ! Разбиение линии № n_line на 1
деление
n_line=n_line+1
*enddo ! i
*do,i,1,2*n ! Цикл по стержням верхнего и нижнего
поясов
LSEL,S,LINE,,n_line ! Выделение линии № n_line
LATT,1,1,1,0 ! Материал №1; константы real 1; эле-
мент LINK1
LESIZE,n_line,,,1 ! Разбиение линии № n_line на 1
деление
n_line=n_line+1
*enddo ! i
*do,i,1,n ! Цикл по стержням раскосов
LSEL,S,LINE,,n_line ! Выделение линии № n_line
LATT,1,2,1,0 ! Материал №1; константы real 2; эле-
мент LINK1
LESIZE,n_line,,,1 ! Разбиение линии № n_line на 1
деление
n_line=n_line+1
*enddo ! i
!
```

## Нелинейные задачи строительной механики

```

*if,del,ne,1,then
LSEL,S,LINE,,n_line ! Выделение линии наружного
троса
LATT,2,3,1,0 ! Материал №2; константы real 3; эле-
мент LINK1
LESIZE,n_line,,,n1 ! Разбиение линии наружного тро-
са на 8 делений
*endif ! i
*if,del,ne,2,then
LSEL,S,LINE,,n_line+1 ! Выделение линии внутреннего
троса
LATT,2,4,1,0 ! Материал №2; константы real 4; эле-
мент LINK1
LESIZE,n_line+1,,,n2 ! Разбиение линии внутреннего
троса на 6 делений
*endif ! i
!
LSEL,S,REAL,,1 ! Выделение линий поясов фермы
LPLOT
LSEL,A,REAL,,2 ! Выделение линий стоек и раскосов
фермы
LPLOT
*if,del,ne,1,then
LSEL,A,REAL,,3 ! Выделение линии наружного троса
LPLOT
*endif ! i
*if,del,ne,2,then
LSEL,A,REAL,,4 ! Выделение линии внутреннего троса
LPLOT
*endif ! i
!
LMESH,ALL ! Построение сетки конечных элементов
!-----
    
```

Вычислительный процесс организуем в рамках динамического анализа поведения ферменно-вантовой конструкции при внезапном отключении одной из тросовых растяжек. Для учета больших перемещений (геометрической нелинейности) используем опцию NLGEOM,1. Этого требует специфика моделирования тросов, для которых необходимо задавать начальные деформации ( $\epsilon_{p1} < 0$ ,  $\epsilon_{p2} < 0$ ), обусловленные предварительным натяжением. Ниже приведен макрос, запускающий соответствующий решатель.

## Нелинейные задачи строительной механики

```

! Расчет предварительно напряженной
! ферменно-вантовой конструкции на прогрессирующее
! разрушение
Pi=4*ATAN(1) ! Число Пи
N=140 ! число шагов по временной координате
FINISH ! Выход из препроцессора
!-----
!Динамический расчет с учетом внезапно приложенного
воздействия
/SOLU ! Вход в процессор решения
!ACEL,0,0,0 ! Удаление гравитационной нагрузки
ANTYPE,4 ! Нестационарное нагружение
TRNOPT,FULL ! "Полный" метод
LUMPM,0
NLGEOM,1 ! Учет больших перемещений
TIMINT,1 ! Метод Ньюмарка
OUTPR,BASIC,ALL, ! Контроль результатов на каждом
шаге
OUTRES,ALL,ALL, ! Вывод результатов на каждом шаге
KBC,0 ! нагрузка от шага к шагу прикладывается
плавно
!
IC,ALL,ALL,0,0, ! Задание нулевых начальных условий
!
*do,I_STEP,1,N,1 ! Цикл по временным шагам
AUTOTS,0 ! Отмена автоматического выбора шага
TIME,I_STEP/(60*Pi) ! Задание текущего шага по вре-
менной оси
*if,I_STEP,eq,1,then
ACEL,0,9.81,0, ! Нагружение фермы собственным ве-
сом
*endif ! i
!
SOLVE ! Запуск решателя
!
*enddo ! Конец цикла по I_STEP
!
FINISH ! Выход из решателя
!-----
    
```

На рис. 22 и 23 представлены графики амплитуд колебаний узла  $a$  (рис. 35) в направлении оси  $Y$  соответственно для вариантов  $del=1$  (отключен наружный трос) и  $del=2$  (отключен внутренний трос).

Для сравнения при статическом расчете величины перемещения узловой точки  $a$  составили: для  $del=1$   $W_y = 0,06003$  м; для  $del=2$   $W_y = 0,07445$  м. Таким образом, ситуация, связанная с отключением одного из тросов, является запроектной в плане динамического отклика стержневой системы.




Рис. 22




Рис. 23

Покадровые картины визуализации колебательного процесса ферменно-вантовой конструкции при  $del=1$  для временных шагов  $i\_step = 10$  и  $i\_step = 16$  приведены на рис. 24. На рис. 25 показаны соответствующие эпюры продольных сил.




Рис. 24




Рис. 25

Расчеты показали, что при  $del=1$  на протяжении всего времени наблюдения за колебательным процессом тросовая растяжка работает на растяжение.

Визуализация колебательного процесса при  $del=2$  для временных шагов  $i\_step = 10$  и  $i\_step = 16$  приведена на рис. 26. Эпюры продольных сил, соответствующие схеме  $del=2$ , представлены на рис. 27.




Рис. 26




Рис. 27



В отличие от схемы  $del=1$  в последнем случае в тросовой растяжке на шаге  $i\_step=16$  возникают сжимающие усилия, что противоречит физическому смыслу задачи, т. к. трос – это гибкая нить, способная воспринимать только растягивающие напряжения. Важно отметить, что эффект смены знака внутреннего усилия в тросе вызывает сбой нелинейного решателя в виде аварийного останова. Это обстоятельство накладывает ограничение на выбор числа шагов  $N$  по временной координате. В частности в рассматриваемом примере для схемы  $del=2$  принималось  $N=40$  (для сравнения для схемы  $del=1$  –  $N=140$ ). Резюмируя можно сделать вывод, что на практике в случае обрыва внутреннего троса и реализации подвеса конструкции по схеме  $del=2$  наружный трос при колебательном процессе будет испытывать резкие рывки при перекладке фермы сверху-вниз.

## ЛИТЕРАТУРА

1. ANSYS: справочник пользователя.-М.:ДМКПресс,2005.-640с.
2. ANSYS в примерах и задач/Под общ. ред. Д.Г.Красковского-М.:КомпьютерПресс,2002.-224с.
3. Бате К., Вильсон Е. Численные методы анализа и метод конечных элементов. М.: Стройиздат, 1982. 447 с.